
HARALD NAHRSTEDT

Excel + VBA

Ergänzungen

Kapitel

Einführung in VBA

Namen in VBA nutzen

Erstellt am 15.10.2011

Beschreibung Eine der großen Stärken von Excel ist die Vergabe von Namen für Zellen und Zellbereiche. Doch Namen sind auch Objekte, deren Attribute und Methoden sich in VBA nutzen lassen. Dieses Kapitel beschreibt den Umgang mit Namen unter VBA.

Eine der großen Stärken von Excel ist die Vergabe von Bereichsnamen für Zellen und Zellbereiche. Doch Bereichsnamen sind auch Objekte, deren Attribute und Methoden sich in VBA nutzen lassen.

1 Die Objektliste Names

Namen stehen als Objektlisten des Workbooks zur Verfügung. Die nachfolgende Prozedur zeigt alle vorhandenen Namen eines Excel-Workbooks.

```
Sub ZeigeAlleNamen()
    Dim objName As Object
    Dim sText As String

    sText = ""
    For Each objName In ThisWorkbook.Names
        sText = sText & vbCrLf & objName.Name
    Next
    MsgBox sText, vbOKOnly, "Vorhandene Namen"
End Sub
```

2 Namen erstellen und löschen

Die Erstellung eines Namens ist nicht an einen Zellbereich gebunden. Im nachfolgenden Beispiel wird ein Name *MehrwertSteuerProzent* erstellt, und ihm wird der Wert 19% zugewiesen. Ab Version 2007 kann auch noch ein Kommentar gesetzt werden.

```
Sub NameErstellenOhneBezug()
    ActiveWorkbook.Names.Add _
    Name:="MehrwertSteuerProzent", RefersToR1C1:="=19%"
    ActiveWorkbook.Names("MehrwertSteuerProzent").Comment = _
    "MehrwertSteuerProzent ist ein Name mit einem" & _
    " festen Wert (ohne Zellbezug)"
End Sub
```

Mit diesem Wert lässt sich dann in der ganzen Mappe rechnen.

	H	I	J	K
4				
5				
6		Betrag	250,00 €	
7		MWSt	=J6*MehrwertSteuerProzent	

	H	I	J
4			
5			
6		Betrag	250,00 €
7		MWSt	47,50 €

Das Löschen eines Namens zeigt die nachfolgende Prozedur.

```
Sub NameLöschen()
    ActiveWorkbook.Names("MehrwertSteuerProzent").Delete
End Sub
```

Der häufigere Fall aber ist einen Namen mit Zellbezug anzulegen. Die nachfolgende Prozedur ordnet dem Zellbereich B2:D12 in der Tabelle1 den Namen *Daten1* zu.

```
Sub NameErstellenMitBezug()  
    ActiveWorkbook.Names.Add Name:="Daten1", _  
        RefersToR1C1:="=Tabelle1!R2C2:R12C4"  
End Sub
```

Eine andere Möglichkeit ist die nachfolgende Form.

```
Sub NameErstellenMitBezug2()  
    Sheets("Tabelle1").Select  
    Range("E3:H14").Name = "Daten2"  
End Sub
```

3 Bereichs- und Namensobjekte

Auch einem Zellbereichsobjekt lässt sich ein Namensobjekt zuordnen.

```
Sub NameErstellenFürBereichsObjekt()  
    Dim Bereichsobjekt As Object  
  
    Set Bereichsobjekt = _  
        ActiveWorkbook.Sheets("Tabelle1").Range("B10:D15")  
    ActiveWorkbook.Names.Add Name:="Daten3", _  
        RefersTo:="=" & Bereichsobjekt.Address(External:=True)  
End Sub
```

Umgekehrt lässt sich ein Namensobjekt einem Zellbereichsobjekt zuordnen.

```
Sub NamenEinemObjektZuordnen()  
    Dim Bereichsobjekt As Object  
  
    Set Bereichsobjekt = _  
        Evaluate(ActiveWorkbook.Names("Daten3").Value)  
End Sub
```

4 Namen darstellen

Der Bereich, dem ein Name zugeordnet ist, lässt sich aktivieren, so dass dieser auf dem Tabellenblatt gekennzeichnet wird.

```
Sub NamensbereichAktivieren()  
    Application.Goto Reference:= _  
        ActiveWorkbook.Names("Daten3").Name  
End Sub
```

Der Bereich, dem ein Name zugeordnet ist, lässt sich in der linken oberen Ecke Ihres Tabellenblattes positionieren.

```
Sub NamensbereichPositionieren()  
    Application.Goto Reference:=  
        ActiveWorkbook.Names("Daten3").Name, Scroll:=True  
End Sub
```

5 Namenbereiche mit Werten füllen

Wie alle Zellen eines Namensbereichs gefüllt werden, zeigt die nachfolgende Prozedur. Darin werden alle Zellen des Bereichs mit dem Wert 123 gefüllt.

```
Sub NamensbereichAusfüllen1()  
    Application.Goto Reference:=ActiveWorkbook. _  
        Names("Daten3").Name  
    Selection.Value = 123  
End Sub
```

Einen Namensbereich ohne Aktivierung auszufüllen, schafft die nächste Prozedur. Hier wird der Text xyz in alle Zellen geschrieben.

```
Sub NamensbereichAusfüllen2()  
    Evaluate(ActiveWorkbook.Names("Daten3").Value).Value = "xyz"  
End Sub
```

Die nachfolgende Prozedur schreibt in alle Zellen des Namensbereich Zufallszahlen.

```
Sub NamensbereichAusfüllen3()  
    Dim Zelle As Object  
    Dim r As Double  
  
    Randomize  
    r = Time  
    For Each Zelle In _  
        ActiveWorkbook.Names("Daten3").RefersToRange.Cells  
        r = Rnd(r)  
        Zelle = r  
    Next  
End Sub
```

Aber auch einzelne Zellen eines Namensbereichs lassen sich beschreiben. In der nachfolgenden Prozedur wird in die erste Zeile und zweite Spalte des Namensbereichs der Wert 33 geschrieben.

```
Sub NamenbereichAusfüllen4()  
    ActiveWorkbook.Names("Daten3").RefersToRange.Cells(1, 2) = 33  
End Sub
```